

Les types utilisateurs : exercices résolus en VBA

Corrigé

Objectifs

- Connaître les types utilisateurs : tableaux, enregistrements et type énumérés ;
- Connaître les algorithmes fondamentaux sur les tableaux ;
- Continuer à appliquer les principes de la semaine 1.

| | |
|---|---|
| Exercice 1 : Occurrences des chiffres d'un entier | 1 |
| Exercice 2 : Modéliser un robot de type 1 | 3 |
| Exercice 3 : Lecture d'un tableau | 7 |
| Exercice 4 : Tri par insertion séquentielle | 8 |

Exercice 1 : Occurrences des chiffres d'un entier

Écrire un programme qui compte le nombre d'occurrences des 10 chiffres dans un entier naturel donné.

Par exemple, l'entier 4214 a une occurrence du chiffre 1, une de 2 et deux de 4.

Solution :

1 **R0** : Compter le nombre d'occurrences des chiffres d'un nombre

Puisque l'objectif est de compter le nombre d'occurrences de chacun des chiffres, l'idée est donc de prendre autant de compteur que de chiffre. Dans ce cas, on prend un tableau de compteur indicé par les chiffres.

```
1 nb: Tableau [0..9] De Entier
2   -- nb[i] = nb d'occurrences de i dans le nombre.
```

On peut alors en déduire l'algorithme suivant :

```
1 R1 : Raffinage De « Compter le nombre d'occurrences des chiffres d'un nombre »
2   | Initialiser les compteurs à 0
3   | Répéter
4   |   | Comptabiliser l'unité de nombre
5   |   | Supprimer l'unité de nombre
6   | JusquÀ nombre = 0
7
8
9 R2 : Raffinage De « Initialiser les compteurs à 0 »
10  | Pour i <- 0 JusquÀ i = 9 Faire
11  |   nb[i] <- 0
12  | FinPour
13
14 R2 : Raffinage De « Comptabiliser l'unité de nombre »
```

```
15 | chiffre = nombre Mod 10
16 | nb[chiffre] <- nb[chiffre] + 1
17
18 R2 : Raffinage De « Supprimer l'unité de nombre »
19 | nombre <- nombre Div 10

1 Attribute VB_Name = "Exo1_compter_chiffres"
2 '*****
3 '* Auteur : Claude Monteil <monteil@ensat.fr>
4 '* Version : 1.0
5 '* Objectif : Compter le nb d'occurrences des chiffres d'un nombre
6 '*****
7
8 Option Explicit
9
10 Sub compter_occurrences()
11     Dim nombre As Long ' entier saisi par l'utilisateur
12     ' (le type Long peut contenir des entiers plus grand que le type Integer)
13     Dim nb(9) As Integer ' nb(i) : nb d'occurrences du chiffre i
14     ' dans nombre (i part de 0 jusqu'à 9)
15     Dim i As Integer ' pour parcourir les chiffres
16     Dim chiffre As Integer ' chiffre unite de nombre
17
18     EffacerEcran "Comptage_des_occurrences_des_chiffres_d'un_nombre"
19
20     '1.saisir le nombre
21     Afficher "Nombre:_:"
22     Saisir nombre
23
24     '2.initialiser les compteurs
25     For i = 0 To 9: nb(i) = 0: Next i
26
27     '3.comptabiliser chaque chiffre de nombre
28     Do 'comptabiliser le chiffre des unites et le supprimer du nombre
29     '3.1.comptabiliser l'unité de nombre
30     chiffre = nombre Mod 10
31     nb(chiffre) = nb(chiffre) + 1
32     '3.2.supprimer l'unité de nombre
33     nombre = nombre \ 10
34
35     Loop Until nombre = 0
36
37     '4.afficher les occurrences
38     For i = 0 To 9
39     Afficher "nb_de_(_" & i & ")_=" & nb(i)
40     Next i
41 End Sub
```

Exercice 2 : Modéliser un robot de type 1

L'objectif de cet exercice est de modéliser un robot de type 1. Un tel robot se déplace dans un environnement qui peut être modélisé par un quadrillage dont chaque case correspond à une position possible du robot.

Un robot est donc caractérisé par sa position (son abscisse, x , et son ordonnée, y) et sa direction (nord, sud, est ou ouest). La figure 1 décrit un robot à la position $x = 4$ et $y = 2$, sa direction est « ouest ».




FIGURE 1 – Le robot dans un environnement illimité

Les robots de type 1 ont des possibilités réduites qui se limitent à pivoter de 90° vers la droite et à avancer d'une case suivant sa direction courante.

Pour simplifier, on considère que le robot évolue dans un environnement infini et sans obstacle.

2.1 Définir les types nécessaires pour modéliser un robot de type 1.

Solution : Un robot de type 1 est caractérisé par :

- une position, c'est-à-dire une abscisse (x) et une ordonnée. On peut donc définir un type **Position** comme un enregistrement.
- une direction. La direction peut prendre 4 valeurs que l'on peut par exemple appeler Nord, Sud, Est et Ouest. On aurait également pu les appeler Gauche, Droite, Haut et Bas. En conséquence, il s'agit d'un type énuméré.

Le type **Robot1** est donc un enregistrement qui regroupe ces deux informations.

```

1  Type
2      Position =
3          Enregistrement
4              x: Entier    -- abscisse
5              y: Entier    -- ordonnée
6          FinEnregistrement
7
8      Direction = (NORD, SUD, EST, OUEST)
9
10     Robot1 =
11         Enregistrement
12             position: Position
13             direction: Direction
14         FinEnregistrement

```

2.2 L'environnement dans lequel évolue le robot est en fait fini et comporte des obstacles. On suppose qu'un obstacle occupe une case du quadrillage. La question est donc de savoir si la case est libre ou contient un obstacle.

Définir un type pour modéliser l'environnement.

Solution : L'environnement dans lequel évolue le robot est un quadrillage, donc il peut être modélisé par un tableau à deux dimensions. Nous utilisons deux constantes pour en préciser les dimensions (NB_COLONNES et NB_LIGNES).

Chaque case du tableau est soit une position qui peut être occupé par le robot, soit un mur, donc deux valeurs possibles. On peut donc représenter cette information par un type booléen (il faut choisir ce que signifie **VRAI** et **FAUX**) ou définir un type énuméré.

Nous choisissons ici cette deuxième solution :

```

1  Constante
2      NB_COLONNES = 20
3      NB_LIGNES = 30
4
5  Type
6      Case = (LIBRE, MUR)
7      -- indique si la case est LIBRE et peut donc être occupé par le robot
8      -- ou si elle contient un mur
9
10     Environnement = Tableau [1..NB_COLONNES, 1..NB_LIGNES] De Case

```

2.3 Écrire un programme qui fait avancer un robot toujours tout droit jusqu'à ce qu'il rencontre un obstacle ou qu'il arrive à la limite de son environnement. On supposera que le robot est initialement sur une position valide avec pour direction « est ».

Solution : On considère un robot particulier dans un environnement donné. Dans les programmes de test, il s'agira donc d'initialiser ces deux informations de manière à ce que l'on puisse tester notre algorithme dans des cas significatifs. Notons que la direction est fixé à l'EST.

```

1  R0 : Faire avancer le robot au maximum
2      robot: in Robot1, environnement: in Environnement
3
4  tests :
5      - pas de mur => la limite est atteinte
6      - un mur => s'arrête devant le mur
7
8  R1 : Raffinage De « Faire avancer le robot au maximum »
9      TantQue (limite environnement non atteinte) et (Pas de mur) Faire
10     Faire avancer le robot d'une case (vers l'est)
11     FinTQ
12
13 R2 : Raffinage De « limite environnement non atteinte »
14     Résultat <- robot.position.x < NB_COLONNES
15
16 R2 : Raffinage De « Pas de mur »
17     Résultat <- environnement[robot.position.x + 1, robot.position.y] <> MUR
18
19 R2 : Raffinage De « Faire avancer le robot d'une case »
20     robot.position.x <- robot.position.x + 1

1  Attribute VB_Name = "Exo2_deplacer_robot"
2  ' *****
3  ' * Auteur : Claude Monteil <monteil@ensat.fr>

```

```

4 ' * Version : 1.1
5 ' * Objectif : Faire avancer un robot vers l'est au maximum
6 ' *****
7
8 Option Explicit
9
10 Const NORD = 0, SUD = 1, EST = 2, OUEST = 3 'directions possibles
11 Const LIBRE = 0, MUR = 1 'indicateur d'occupation d'une cellule
12
13 Type Posi ' une position
14     x As Integer ' abscisse
15     y As Integer ' ordonnée
16 End Type
17
18 Type Robot1
19     position As Posi
20     direction As Integer ' parmi les 4 constantes NORD, SUD, EST, OUEST
21 End Type
22
23
24 Sub deplacer_robot()
25
26     Const NB_LIGNES = 20, NB_COLONNES = 30
27     Dim environnement(NB_LIGNES, NB_COLONNES) As Integer
28     Dim robot As Robot1
29     Dim l, c As Integer ' parcourir les lignes et colonnes de l'environnement
30
31     EffacerEcran "Déplacement_d'un_robot"
32
33     '1.Initialiser l'environnement
34     Afficher "Initialiser_l'environnement..."
35     For l = 1 To NB_LIGNES
36         For c = 1 To NB_COLONNES
37             environnement(l, c) = LIBRE
38         Next c
39     Next l
40     ' Mettre un mur en abscisse 10
41     For c = 1 To NB_COLONNES
42         environnement(10, c) = MUR
43     Next c
44
45     '2.Initialiser le robot
46     Afficher "Initialiser_le_robot..."
47     robot.direction = EST
48     robot.position.x = 1
49     robot.position.y = 1
50
51     '3.Faire avancer le robot
52     Do Until (robot.position.x = NB_LIGNES) _
53         Or environnement(robot.position.x + 1, robot.position.y) = MUR
54         '3.1.faire avancer le robot d'une case (vers l'est)
55         robot.position.x = robot.position.x + 1

```

```
56         '3.2.Afficher la position du robot
57         Afficher "Le_robot_est_en:_:" & robot.position.x & "," & robot.position.y
58     Loop
59 End Sub
```

Exercice 3 : Lecture d'un tableau

Écrire un programme qui initialise un tableau d'entiers en lisant des valeurs au clavier. On considèrera que les valeurs sont saisies les unes après les autres et que la saisie s'arrête sur une valeur négative. La valeur négative ne doit pas être conservée dans le tableau.

Solution :

```

1  Attribute VB_Name = "Exo3_lecture_tableau"
2  '*****
3  '* Auteur   : Claude Monteil <monteil@ensat.fr>
4  '* Version  : 1.0
5  '* Objectif :
6  '*         Saisir un tableau en s'arrêtant sur une valeur negative.
7  '*****
8
9  Option Explicit
10
11 Sub lecture_tableau()
12     Const CAPACITE As Integer = 10     ' capacite du tableau a lire
13     Dim tableau(CAPACITE) As Integer   ' la tableau
14     Dim taille As Integer              ' la taille effective du tableau
15     Dim valeur As Integer              ' valeur lue au clavier
16     Dim i As Integer                  ' pour parcourir les indices du tableau
17
18     EffacerEcran "Lecture_d'un_tableau"
19
20     '1.Saisir le tableau
21     Afficher "Entrez_les_composantes_du_tableau"
22     Afficher "Entrez_une_valeur_negative_pour_arreter."
23     taille = 0 '
24     Saisir valeur
25     Do While (valeur >= 0) And (taille < CAPACITE)
26         ' ranger la valeur
27         taille = taille + 1
28         tableau(taille) = valeur
29         ' Saisir une nouvelle valeur
30         If taille < CAPACITE Then Saisir valeur
31     Loop
32     If (taille = CAPACITE) Then
33         Afficher "Capacite_du_tableau_atteinte._Arret_de_la_saisie"
34     End If
35
36     '2.Afficher le tableau (verticalement)
37     For i = 1 To taille
38         Afficher tableau(i)
39     Next i
40 End Sub

```

Exercice 4 : Tri par insertion séquentielle

Soit $A[1..N]$ un vecteur de N entiers relatifs quelconques, l'objectif est de trier le vecteur A . Le vecteur A est trié si $A[1] \leq A[2] \leq \dots \leq A[N]$.

Le tri utilisé est le tri par insertion séquentielle. C'est un tri en $(N - 1)$ étapes. L'étape i consiste à placer le $(i + 1)^{\text{e}}$ élément du vecteur à sa place dans le sous-vecteur $A(1..i + 1)$ sachant que $A(1..i)$ a été trié par les étapes précédentes. Dans le cas du tri par insertion séquentielle, la recherche de la position d'insertion, se fait séquentiellement en comparant successivement l'élément à insérer aux i premiers éléments du vecteur.

Exemple : Voici les différentes valeurs du vecteur 8 2 9 5 1 7 après chaque étape (la partie encadrée correspond à la partie du vecteur déjà traitée et donc triée) :

| | | |
|-----------------|---|---|
| vecteur initial | : | 8 2 9 5 1 7 |
| après l'étape 1 | : | 2 8 9 5 1 7 |
| après l'étape 2 | : | 2 8 9 5 1 7 |
| après l'étape 3 | : | 2 5 8 9 1 7 |
| après l'étape 4 | : | 1 2 5 8 9 7 |
| après l'étape 5 | : | 1 2 5 7 8 9 |

4.1 Écrire un programme qui lit une série de N ($N > 0$, lu au clavier) nombres relatifs, les range dans un vecteur A , les classe par ordre croissant en utilisant le tri par insertion séquentielle et, enfin, affiche la série ordonnée.

Solution :

1 **R0** : Trier un tableau A de taille effective N (les indices sont $1..N$).

On remarque qu'après chaque étape i du tri, les $(i+1)$ premiers éléments sont à leur place. Plus précisément, le sous-tableau compris entre les indices 1 et $i+1$ est trié. Ainsi, au bout de la $(N-1)^{\text{e}}$ étape l'ensemble du tableau est trié.

L'étape i consiste à trier le tableau $1..i+1$ en sachant que le tableau $1..i$ est déjà trié.

1 **R1** : **Raffinage De** << Trier un tableau A de taille effective N >>

2 | **Pour** indice := 2 **JusquÀ** indice = N **Faire**

3 | Insérer $A[\text{indice}]$ dans $A(1..\text{indice})$ sachant que $A(1..\text{indice})$ est trié

4 | **FinPour**

5

6 **R2** : **Raffinage De** << Insérer $A[\text{indice}]$ dans $A(1..\text{indice})$ sachant que $A(1..\text{indice})$ est trié >>

7 | Déterminer la position théorique de $A[\text{indice}]$ dans $A(1..\text{indice})$

8 | Décaler les éléments compris entre la position théorique et $\text{indice}-1$

9 | Ranger l'élément $A[\text{indice}]$

Si on essaie d'exécuter mentalement ce raffinement¹, on constate que l'on a perdu la valeur $A[\text{indice}]$ lorsque l'on veut la ranger. Elle a été écrasée lors du raffinement. Il est donc nécessaire de la conserver avant et donc de revoir notre raffinement.

1 **R2** : **Raffinage De** << Insérer $A[\text{indice}]$ dans $A(1..\text{indice})$ sachant que $A(1..\text{indice})$ est trié >>

2 | Conserver la valeur de $A[\text{indice}]$ memoire: **out Entier**

1. Ce qui doit toujours être fait, puisque ceci fait partie des choses à faire pour vérifier qu'un raffinement est correct.


```

3   | Déterminer la position pos de A[indice] dans A(1..indice)
4   | Décaler les éléments compris entre pos et indice-1
5   | Ranger l'élément mémorisé
6
7   R3 : Raffinage De << Déterminer la position théorique de A[indice] dans A(1..indice) >
8   | pos <- 1
9   | TantQue (pos < indice ) Et (memoire >= A[pos]) Faire
10  |     pos <- pos + 1
11  | FinTQ
12
13  R3 : Raffinage De << Décaler les éléments compris entre la position théorique et indice
14  | Pour i <- indice - 1 Décrémenter Jusqu'À i = pos Faire
15  |     A[i+1] <- A[i]
16  | FinPour

```

On peut en déduire l'algorithme.

```

1   Constante
2     CAPACITÉ = 100
3   Type
4     Vecteur = Tableau [1..CAPACITÉ] De Entier
5
6   Variable
7     A: Vecteur -- le tableau à trier
8     N: Entier -- le nombre d'éléments de A
9     indice: Entier -- l'indice de l'élément à insérer
10    memoire: Entier -- mémoriser A[indice]
11    pos: Entier -- position de l'élément mémoire dans A(1..indice)
12    i: Integer; -- parcourir les éléments du tableau
13
14  Début
15    Pour indice := 2 Jusqu'À indice = N Faire
16      -- Insérer A[indice] dans A(1..indice) sachant que A(1..indice) est trié
17      -- Conserver la valeur de A[indice]
18      memoire <- A[indice]
19
20      -- Déterminer la position pos de A[indice] dans A(1..indice)
21      pos <- 1
22      TantQue (pos < indice ) Et (memoire >= A[pos]) Faire
23        pos <- pos + 1
24      FinTQ
25
26      -- Décaler les éléments compris entre pos et indice-1
27      Pour i <- indice - 1 Décrémenter Jusqu'À i = pos Faire
28        A[i+1] <- A[i]
29      FinPour
30
31      -- Ranger l'élément mémorisé
32      A[pos] <- memoire
33    FinPour
34  Fin

```

On peut en écrire une version légèrement optimisée en réalisant les décalages en même temps que l'on cherche la position d'insertion.

```

1  Variable
2      A: Vecteur  -- le tableau à trier
3      N: Entier   -- le nombre d'éléments de A
4      indice: Entier   -- l'indice de l'élément à insérer
5      mémoire: Entier  -- mémoriser A[indice]
6      pos: Entier  -- pour chercher la position d'insertion dans A(1..indice)
7
8  Début
9      Pour indice := 2 JusquÀ indice = N Faire
10         -- Insérer A[indice] dans A(1..indice) sachant que A(1..indice) est trié
11         -- Conserver la valeur de A[indice]
12         mémoire <- A[indice]
13
14         -- Faire de la place pour A[indice] dans A(1..indice)
15         pos <- indice
16         TantQue (pos > 1 ) Et (memoire < A[pos-1]) Faire
17             A[pos] <- A[pos-1]
18             pos <- pos - 1
19         FinTQ
20
21         -- Ranger l'élément mémorisé
22         A[pos] <- mémoire
23     FinPour
24 Fin

1  Attribute VB_Name = "Exo4_tri_par_insertion"
2  '*****
3  '* Auteur   : Claude Monteil <monteil@ensat.fr>
4  '* Version  : 1.0
5  '* Objectif : Trier un tableau par insertion séquentielle
6  '*****
7
8  Option Explicit
9
10 Sub trier_par_insertion()
11     Const MAX As Integer = 20 ' Capacité maximale du tableau
12     Dim tableau(MAX) As Integer ' la tableau a trier
13     Dim nb As Integer ' la taille effective du tableau a trier
14     Dim valeur As Integer ' valeur lue au clavier
15     Dim indice As Integer ' tableau(0..indice) est trie
16     Dim position As Integer ' position d'insertion de tableau(indice)
17     Dim i As Integer ' variable de boucle
18     Dim memoire As Integer ' conserver la valeur de tableau(indice)
19
20     EffacerEcran "Tri_d'un_tableau_par_insertions_successives"
21
22     '1.Saisir le tableau
23     nb = 0 'nombre de valeurs lues
24     Afficher "Entrez_une_valeur_negative_pour_arreter."
25     Saisir valeur

```

```

26     Do While (valeur >= 0) And (nb < MAX)
27         ' ranger la valeur
28         nb = nb + 1
29         tableau(nb) = valeur
30         ' Saisir une nouvelle valeur
31         If nb < MAX Then Saisir valeur
32     Loop
33     If (nb = MAX) Then
34         Afficher "Capactite_du_tableau_atteinte. Arret_de_la_saisie"
35     End If
36
37     '2.Trier le tableau
38     For indice = 1 To nb ' Insérer tableau(indice) dans tableau(1..indice)
39
40         '2.1.Conserver la valeur de tableau(indice)
41         memoire = tableau(indice)
42         '2.2.determiner la position theorique de tableau(indice)
43         position = 1
44         Do While (position < indice) And (memoire >= tableau(position))
45             position = position + 1
46         Loop
47         '2.3.decaler les elements compris entre position et indice-1
48         For i = indice - 1 To position Step -1
49             ' le pas est negatif : parcours en sens inverse
50             tableau(i + 1) = tableau(i)
51         Next i
52         '2.4.ranger l'element
53         tableau(position) = memoire
54     Next indice
55
56     '3.Afficher le tableau (verticalement)
57     Afficher "Tableau_trié:"
58     For i = 1 To nb
59         Afficher tableau(i)
60     Next i
61 End Sub

```

4.2 En conservant le principe du tri par insertion, expliquer comment améliorer l'efficacité de cet algorithme.

Solution : Si on respecte le principe du tri par insertion, il faudra toujours réaliser les décalages. On ne peut donc gagner en performance qu'en optimisant la recherche (l'optimisation présentée précédemment n'est pas une optimisation réellement intéressante !). Comme on sait que le tableau $A[1..indice-1]$ est trié, on peut utiliser une recherche par dichotomie pour rechercher la position théorique de l'élément dans le tableau.